Question: Is it possible to change the Dashed Style of a curve into circles with Plot command ?

```
Plot[Cos[x], {x, 0, 2*Pi}, PlotStyle -> Directive[Dashed, Thickness[0.005]]]

```

![enter image description here](https://i.stack.imgur.com/Q1554.jpg)

Answer: 
Perhaps this approach, using [`Mesh`](http://reference.wolfram.com/mathematica/ref/Mesh.html), [`MeshStyle`](http://reference.wolfram.com/mathematica/ref/MeshStyle.html), and [`Opacity`](http://reference.wolfram.com/mathematica/ref/Opacity.html), will give you what you looking for.

```
Plot[Cos[x], {x, 0, 2 Pi},
  PlotStyle -> {Opacity[0]}, Mesh -> 50, MeshStyle -> {PointSize[Medium]}]

```

![plot.png](https://i.stack.imgur.com/NxB9h.png)

### Update

To show two curves, one as dots and the other as a normal plot, I would make two plots and combine them with [`Show`](http://reference.wolfram.com/mathematica/ref/Show.html).

```
p1 = 
  Plot[Cos[x], {x, 0, 2 Pi}, 
   PlotStyle -> {Opacity[0]}, Mesh -> 30, MeshStyle -> {PointSize[Large], Blue}];
p2 = Plot[Sin[x], {x, 0, 2 Pi}, PlotStyle -> {Thick, Blue}];
Show[{p1, p2}]

```

![show.png](https://i.stack.imgur.com/3RzdU.png)

You can use `Show` to combine as many plots as you wish.
This is an alternate compute-your-own-arc-length approach, based directly on the line produced by plot, (ie not differentiating its functional form as in Mark's answer)

```
p1 = Plot[Cos[x], {x, 0, 2 Pi}]
linepoints[plist_, n_] := Module[{arc, tlen},
  arc = Interpolation[
        Transpose@{{0}~Join~
                  Accumulate@(Norm@(Subtract @@ #) & /@ 
                       Partition[plist, 2, 1]), plist},
        InterpolationOrder -> 3];
  tlen = Last@First@First@(arc);
  Table[ arc[tlen iz / n], {iz, 0, n}]];
ListPlot[
       linepoints[
          First@(First@Cases[FullForm[ p1], Line[__], Infinity]),  25],
       PlotMarkers -> ({Graphics@Circle[{0, 0}], .025})]

```

![enter image description here](https://i.stack.imgur.com/Sxvxh.png)

A nice illustration comparing equal spaced points vs a more simple ListPlot[Table]] approach:

```
p1 = Plot[Cos[x^2], {x, 0, Pi}];
Show[
 {ListPlot[
     linepoints[First@(First@Cases[FullForm[ p1], Line[__], Infinity]), 
     120], PlotMarkers -> ({Graphics@Circle[{0, 0}], .025})],
  ListPlot[Table[{x, -Cos[x^2]}, {x, 0, Pi, 2 Pi/120}], 
          PlotMarkers -> ({Graphics@Rectangle[{0, 0}], .025})]}, 
    PlotRange -> All]

```

![enter image description here](https://i.stack.imgur.com/K1s14.png)